

www.desarrollosocial.cl

Pobreza Multidimensional con Entorno y Redes: Anexo Programación de Indicadores y Medición

Serie Documentos Metodológicos Casen N° 33
9 enero 2017

Casen 2015

Encuesta de Caracterización
Socioeconómica Nacional

TABLA DE CONTENIDOS

1.	Introducción	3
2.	Uso del programa para a generación de indicadores y medida de pobreza multidimensional.....	3
3.	Sintaxis de programación en software Stata	6



1. Introducción

En el marco de la difusión de la encuesta de Caracterización Socioeconómica Nacional Casen 2015, en complemento al documento “Metodología de medición de pobreza multidimensional con entorno y redes”, se pone a disposición de los usuarios la sintaxis de programación de la medida ampliada de Pobreza Multidimensional, presentada por el Ministerio de Desarrollo Social en agosto 2016, que incluye 5 dimensiones de bienestar: Educación, Salud, Trabajo y Seguridad Social, Vivienda y Entorno, y Redes y Cohesión Social. Esta sintaxis permite identificar los hogares carentes en cada indicador de las 6 dimensiones y luego identificar los hogares y personas en situación de pobreza multidimensional. Cabe mencionar que esta sintaxis es aplicable sólo a la versión 2015 de la encuesta, pues que en el cuestionario Casen 2015 se incorporaron nuevas preguntas que permitieron construir los indicadores de Entorno, Redes y apoyo Social, Seguridad, y Trato Igualitario.

De esta forma, la sintaxis informada permite construir los indicadores que conforman las 5 dimensiones usando la base de datos de la Encuesta Casen 2015 y luego ejecutar los pasos posteriores de identificación, agregación y creación de los índices de pobreza multidimensional (h, a y m0) establecidos en el marco de la metodología.

También se incluyen las instrucciones necesarias para construir la medición con cuatro dimensiones, publicada en enero 2015, cuyos resultados son comparables con la serie de datos presentada por el Ministerio de Desarrollo Social para los años 2009, 2011 y 2013.

Tanto la base de datos, como el cuestionario y libro de códigos de la Encuesta Casen 2015 se encuentran publicados en el sitio web ministerial¹. Mayores detalles de la metodología de pobreza multidimensional pueden encontrarse en documento metodológico antes señalado².

2. Uso del programa para a generación de indicadores y medida de pobreza multidimensional

Para generar la medida de pobreza multidimensional se entrega en la sección 3 la sintaxis de programación en formato STATA (Do-File), denominado “**Indicadores de Pobreza Multidimensional**”. Esta permite generar los indicadores que se utilizan para identificar carencias en las dimensiones de Educación, Salud, Trabajo y Seguridad Social, Vivienda y Entorno, y Redes y Cohesión Social.

Dichas dimensiones son agregadas para obtener el índice de pobreza multidimensional usando la base datos de la Encuesta Casen 2015.

¹ Disponible en: <http://observatorio.mds.cl/casen-multidimensional/casen/basedatos.php> (diciembre, 2016).

² Disponible en: [http://observatorio.mds.cl/casen-multidimensional/casen/docs/Metodologia de Medicion de Pobreza Multidimensional.pdf](http://observatorio.mds.cl/casen-multidimensional/casen/docs/Metodologia%20de%20Medicion%20de%20Pobreza%20Multidimensional.pdf) (diciembre, 2016).

El documento se organiza de la siguiente forma:

0. Preámbulo: En el cual se detalla la base de datos que se deberá utilizar para la generación de los indicadores. Se debe especificar directamente la "Base de Datos Casen 2015"³.

Además, deben excluirse los casos que corresponden a núcleos compuestos por el servicio doméstico (nucleo=0) y fijar el comando correspondiente (*svyset*) para trabajar con el diseño complejo de la encuesta Casen para este año. Adicionalmente, se borran las variables de los indicadores multidimensionales incluidas en la base de datos, para evitar la duplicación en el proceso de su construcción.

1. Variables necesarias para los indicadores de pobreza multidimensional: Este apartado considera la determinación de la aquellas variables requeridas para la construcción de los indicadores referidos a las carencias en cada una de las dimensiones individuales que se consideran (Educación, Salud, Trabajo y Seguridad Social, Vivienda y Entorno y Redes y Cohesión Social).

2. Desagregaciones para las estadísticas descriptivas relativas a los indicadores de pobreza multidimensional.

3. Sintaxis para la creación de los indicadores de pobreza multidimensional con entorno y redes, así como la versión de medida de pobreza con cuatro dimensiones: En este apartado se define la sintaxis que permite crear cada uno de los indicadores de carencias dentro de las dimensiones de educación, salud, trabajo y seguridad social, vivienda y entorno, redes y cohesión social. Para cada indicador, en una primera etapa, se define la población de referencia (es decir, quiénes son susceptibles de ser afectados por la carencia que identifica el indicador). Luego, se establece cuáles son las personas que tienen la carencia en la población de referencia.

Para fines de comparabilidad con los años 2009, 2011 y 2013, se generan también los indicadores de pobreza considerados en la medición con cuatro dimensiones (educación, salud, trabajo y seguridad social y vivienda). Es de resaltar que a partir de Casen 2015, se inicia una serie nueva para medir pobreza con la inclusión de Entorno, Redes y Cohesión Social.

4. Agregación de indicadores por hogar: Se identifican los hogares que presentan carencia en cada uno de los indicadores, de acuerdo a sus poblaciones de referencia.

³ En el 2015 se debe descargar la base llamada "Base de Datos CASEN 2015" en el siguiente link: <http://observatorio.ministeriodesarrollosocial.gob.cl/casen-multidimensional/casen/basedatos.php>.

5. Identificación y Agregación de Medida de Pobreza Multidimensional con Entorno y Redes. Se compone de dos etapas. En la primera de ellas, la identificación, donde se determinan cuáles hogares se clasifican como pobres multidimensionales. Primero se pondera cada una de las carencias, de acuerdo a la actual metodología establecida para Chile se considera que cada dimensión tradicional, estas son, Educación, Salud, Trabajo y Seguridad Social, Vivienda y Entorno, tiene la misma ponderación (22,5%). Y en cuanto a la dimensión de Redes y Cohesión Social tiene un peso de 10%. Dentro de cada dimensión, los indicadores tienen igual peso. Posteriormente se establece el vector de conteo, que corresponde a la suma del valor de cada indicador ponderado. Además se identifican los hogares que no cuentan con información en alguno de los indicadores que componen la medida. Una vez determinado el vector de conteo para los hogares se clasifica la situación de pobreza de un hogar para los hogares que se dispone información completa. El umbral para determinar una situación de pobreza es de $k=22,5\%$, calculando así el indicador de Incidencia de Pobreza Multidimensional ("h").

En la segunda etapa, la agregación, se genera el indicador de intensidad de la pobreza multidimensional ("a"), el cual establece el promedio ponderado de carencias que experimentan los hogares en situación de pobreza. Asimismo en esta etapa se elabora el Índice M0, que corresponde a la multiplicación de la incidencia (h) por la intensidad de la pobreza (a).

6. Identificación y Agregación de Medida de Pobreza Multidimensional con 4 dimensiones. Al igual que el punto 5, se realiza el mismo proceso para estimar la incidencia de la pobreza medida con cuatro dimensiones: Educación, salud, trabajo y seguridad social, y vivienda. El objetivo de este proceso de la programación es establecer para este año comparabilidad con las estimaciones presentadas en el año 2009, 2011 y 2013.

3. Sintaxis de programación en software Stata

```
*****
* Calculo de indicadores y medición de Pobreza Multidimensional Casen 2015
*****
```

```
*****
* 0. Preámbulo
*****
```

```
*****
*****
```

```
clear all
```

```
*Se selecciona la base de datos
```

```
use "C:\...\Casen 2015 STATA.dta", clear /*REEMPLAZAR RUTA SEGÚN CORRESPONDA*/
```

```
* Se eliminan las observaciones correspondientes al núcleo cero
```

```
drop if nucleo==0
```

```
* Muestra Compleja
```

```
svyset varunit [pw=expr], strata(varstrat) singleunit(certainty)
```

```
*Se eliminan las variables multidimensionales previamente cargadas en la base de datos
```

```
local var asis rez esc mal prevs acc act cot jub hacina estado habitab servbas entorno appart tsocial seg  
accesi medio
```

```
foreach var in `var' {
```

```
drop hh_d_`var'
```

```
}
```

```
drop pobreza_multi_4d pobreza_multi_5d
```

```
*****
*****
```

```
* 1. Variables necesarias para la creación de los indicadores
```

```
*****
*****
```

```
* Generar escolaridad para los menores de 15 años
```

```
gen esc_2=esc
```

```
replace esc_2=0 if (esc==. & e6a<=5) & e6a!=99
```

```
replace esc_2=e6b if (esc==. & e6a==7) & e6a!=99
```

```
replace esc_2=e6b+8 if (esc==. & (e6a==9|e6a==11)) & e6a!=99
```

```
replace esc_2=. if (e6a==99|e6b==99) & esc==.
```

```
* Genera Grado para rezago
```

```
gen grado=.
```

```
replace grado=1 if e6b==1 & e6a==7 & e6a!=99 & e6b!=99
```

```
replace grado=2 if e6b==2 & e6a==7 & e6a!=99 & e6b!=99
```

```
replace grado=3 if e6b==3 & e6a==7 & e6a!=99 & e6b!=99
```

```
replace grado=4 if e6b==4 & e6a==7 & e6a!=99 & e6b!=99
```

```
replace grado=5 if e6b==5 & e6a==7 & e6a!=99 & e6b!=99
```

```
replace grado=6 if e6b==6 & e6a==7 & e6a!=99 & e6b!=99
```

```
replace grado=7 if e6b==7 & e6a==7 & e6a!=99 & e6b!=99
```

```
replace grado=8 if e6b==8 & e6a==7 & e6a!=99 & e6b!=99
```

```
replace grado=9 if e6b==1 & (e6a==9|e6a==11) & e6a!=99 & e6b!=99
```

```
replace grado=10 if e6b==2 & (e6a==9|e6a==11) & e6a!=99 & e6b!=99
```

```
replace grado=11 if e6b==3 & (e6a==9|e6a==11) & e6a!=99 & e6b!=99
```

```
replace grado=12 if (e6b==4|e6b==5) & (e6a==9|e6a==11) & e6a!=99 & e6b!=99
```

```

gen curso=grado+6 if asiste==1

* Nivel Educativo *
gen nivel=.
replace nivel=0 if (e6a==1|e6a==2|e6a==3|e6a==4 |e6a==5)
replace nivel=1 if e6a==6 & e6b<6
replace nivel=2 if e6a==6 & e6b==6
replace nivel=1 if e6a==7 & e6b<8
replace nivel=2 if e6a==7 & e6b==8
replace nivel=3 if e6a==8 & e6b<6
replace nivel=5 if e6a==8 & e6b==6
replace nivel=3 if e6a==9 & e6b<4
replace nivel=5 if e6a==9 & e6b==4
replace nivel=4 if e6a==10 & e6b<6
replace nivel=4 if e6a==11 & e6b<5
replace nivel=6 if e6a==10 & (e6b==4|e6b==5|e6b==6)
replace nivel=6 if e6a==11 & e6b==4
replace nivel=7 if e6a==12
replace nivel=8 if e6a==13
replace nivel=9 if e6a==14
replace nivel=10 if e6a==16
replace nivel=11 if e6a==15
replace nivel=12 if e6a==17
replace nivel=99 if (e6a==99|e6b==99)

```

```

*****
*****
* 2. Desagregaciones para estadísticas descriptivas
*****
*****

```

```

* Se generan los tramos etarios de las estadísticas descriptivas
gen t_edad=.
replace t_edad=1 if edad<=18
replace t_edad=2 if edad>18 & edad<35
replace t_edad=3 if edad>34 & edad<65
replace t_edad=4 if edad>64
label define t_edad 1 "0 a 18 años" 2 "19 a 34 años" 3 "35 a 65 años" 4 "65 años o más"
label value t_edad t_edad
label variable t_edad "Tramos etarios"

```

```

* Se generan los tramos para el tamaño del hogar de las estadísticas descriptivas
gen t_hogar=.
replace t_hogar=1 if numper==1
replace t_hogar=2 if numper==2
replace t_hogar=3 if numper==3
replace t_hogar=4 if numper==4
replace t_hogar=5 if numper>4
label define t_hogar 1 "Hogar Unipersonal" 2 "2 integrantes" 3 "3 integrantes" 4 "4 integrantes" 5 "5 o
más integrantes del hogar"
label value t_hogar t_hogar
label variable t_hogar "Tamaño del Hogar"

```


*** 3. Definición de los indicadores del Índice de Pobreza Multidimensional**

*** 3.a. Indicadores de Educación**

 * Indicador 1: Asistencia a un establecimiento educacional
 * La población de referencia corresponde a las personas entre 4 y 18, incluyendo también a quienes tienen entre 6 y 26 años encontrándose en
 * situación de discapacidad.
 * Se considera carente a los hogares en que al menos uno de sus integrantes de la población de referencia no asiste a un establecimiento
 * educacional.
 * Se excluyen de los carentes a las personas que han culminado la enseñanza media, es decir, cuentan con los 12 años de educación obligatoria.

* Población de referencia

```
gen pob_asis=.
replace pob_asis=0 if (edad<4 | edad>18) & edad!=.
replace pob_asis=1 if (edad>=4 & edad<=18) & edad!=.
replace pob_asis=1 if (edad>=6 & edad<=26) & (s31c1<7|s31c2<7|s31c3<7) & s31c1!=9
replace pob_asis=0 if (edad<6) & (s31c1<7|s31c2<7|s31c3<7) & s31c1!=9
```

```
replace pob_asis=0 if (esc_2>=12 & esc_2!=.)
```

```
replace pob_asis=8 if esc_2==. & pob_asis==1 & edad>=15
replace pob_asis=8 if ((edad>=4 & edad<6)|(edad>18 & edad<=26)) & s31c1==9 & esc_2<12
replace pob_asis=8 if pob_asis==.
```

* Incidencia a nivel individual

```
gen d_asis=.
replace d_asis=1 if (asiste==2) & pob_asis==1
replace d_asis=0 if (asiste==1) & pob_asis==1
replace d_asis=9 if d_asis==. & pob_asis==1
```

 * Indicador 2: Rezago Escolar

* La población de referencia incluye a toda persona de 21 años o menos que asiste a educación básica, educación media científico-humanista o
 * educación media técnico profesional.
 * Se considera carentes a aquellos hogares en que al menos uno de sus integrantes de la población de referencia se encuentra retrasado en dos o
 * más años en sus estudios.

*Población de referencia

```
gen pob_rez=.
replace pob_rez=0 if edad>21 & edad!=.
replace pob_rez=0 if asiste==2
replace pob_rez=0 if (e6a!=7 & e6a!=9 & e6a!=11)
```

```
replace pob_rez=1 if (asiste==1 & (e6a==7|e6a==9|e6a==11) & (edad<=21) & (e6a!=99))
replace pob_rez=0 if esc_2>=12 & esc_2!=. & pob_rez==1
```

* Privación a nivel individual

```
gen d_rez=.
replace d_rez=1 if(edad-curso>=2) & pob_rez==1 & asiste!=. & esc_2!=. & edad!=.
replace d_rez=0 if(edad-curso<2) & pob_rez==1 & asiste!=. & esc_2!=. & edad!=.
replace d_rez=9 if d_rez==. & pob_rez==1
```

* Indicador 3: Escolaridad

* La población de referencia corresponde a todas las personas mayores de 18 años, tanto quienes asisten y quienes no lo hacen

* Se considera carente a aquellos hogares que tienen entre sus integrantes a una persona que ha alcanzado menos años de estudios que los

* establecidos por ley (de acuerdo a su edad)

* Se definen exigencias por ley para cada cohorte que se ve afectada por los cambios legislativos

* a. 1920 a 1929: 4 años

* b. 1930 a 1965: 6 años

* c. 1966 a 2002: 8 años

* d. 2003 en adelante: 12 años

* Se generan los tramos etarios, considerando la edad que se debiera tener al entrar en vigencia el cambio de legislación

```
gen tedad=1 if (edad>105) & edad!=.
```

* A quienes les comenzó a regir los 4 años de escolaridad obligatoria en el 2015 tienen 105 años (1920)

```
replace tedad=2 if (edad>97 & edad<=105)
```

* A quienes les comenzó a regir los 6 años de escolaridad obligatoria en el 2015 tienen 97 años (1930)

```
replace tedad=3 if (edad>62 & edad<=97)
```

* A quienes les comenzó a regir los 8 años de escolaridad obligatoria en el 2015 tienen 62 años (1966)

```
replace tedad=4 if (edad>30 & edad<=62)
```

* A quienes les comenzó a regir los 12 años de escolaridad obligatoria en el 2015 tienen 30 años (2003)

```
replace tedad=5 if (edad>18 & edad<=30)
```

* Población de referencia

```
gen pob_esc=.
```

```
replace pob_esc=1 if (edad>18) & edad!=.
```

```
replace pob_esc=0 if (edad<=18) & edad!=.
```

```
replace pob_esc=8 if pob_esc==.
```

* Privación a nivel individual

```
gen d_esc=.
```

```
replace d_esc=0 if ((tedad==1)|((tedad==2) & esc>=4)|((tedad==3 & esc>=6)|((tedad==4 & esc>=8)|((tedad==5 & esc>=12 )) & esc!=. & pob_esc==1
```

```
replace d_esc=1 if ((tedad==2 & esc<4)|((tedad==3 & esc<6)|((tedad==4 & esc<8)|((tedad==5 & esc<12 )) & esc!=. & pob_esc==1
```

```
replace d_esc=9 if d_esc==. & pob_esc==1
```

*** 3.b. Indicadores de Salud**

* Indicador 4: Malnutrición en niños y niñas

* La población de referencia considera a los niños y niñas de 0 a 6 años

* Se considera carentes a los hogares que cuentan con al menos un niño o niña que este desnutrido, en riesgo de desnutrición, con sobrepeso u

* obesidad.

* Población de referencia

gen pob_des= (edad<=6)

gen pob_sob= pob_des

gen pob_mal= pob_des

* Privación a nivel individual

* Para niños y niñas desnutridos o en riesgo de desnutrición

gen d_des=.

replace d_des=1 if(s1==1) & pob_des==1

replace d_des=0 if(s1==2|s1==3|s1==4) & pob_des==1

* Para niños y niñas con sobrepeso u obesidad

gen d_sob=.

replace d_sob=1 if (s1==3|s1==4) & pob_sob==1

replace d_sob=0 if (s1==1|s1==2) & pob_sob==1

* Niños y niñas con Malnutrición

gen d_mal=(d_des==1|d_sob==1) if (d_des!=.|d_sob!=.) & pob_des==1

replace d_mal=9 if s1==9 & pob_des==1

* Indicador 5: Adscripción al Sistema de Salud

* La población de referencia corresponde a todas las personas

* Se considera carentes a los hogares en que al menos uno de sus integrantes que no este afiliado a un sistema previsional de salud y no tiene

* otro seguro complementario.

* Población de referencia

gen pob_prevs=(s12!=.)

* Privación a nivel individual

gen d_prevs=.

replace d_prevs=0 if (s12<8 | s12==9) & pob_prevs==1

replace d_prevs=1 if (s12==8) & pob_prevs==1

replace d_prevs=9 if s12==99 & pob_prevs==1

bys nucleo folio: egen s14_c=min(s14)

replace d_prevs=0 if s14_c==1 & pob_prevs==1

* Indicador 6: Atención de Salud

* La población de referencia corresponde a todas las personas que necesitaron atención médica o están en tratamiento por patología garantizada

* (AUGE-GES)
 * Se considera carentes a los hogares que cuentan con al menos un integrante que no recibió atención de salud en los últimos tres meses o no tuvo
 * cobertura del sistema AUGE-GES por razones ajenas a su voluntad o preferencia.

* Población de referencia
 * Población que no tuvo consulta por enfermedad en 3 meses
 gen pob_acc_a=(s16==2)
 replace pob_acc_a=8 if(s16==9)

* Población en tratamiento de enfermedades auge sin tratamiento por AUGE
 gen pob_acc_b=(s29==2)
 replace pob_acc_b=8 if(s29==9)

* Población de referencia total
 gen pob_acc=(pob_acc_a==1|pob_acc_b==1)
 replace pob_acc=8 if (pob_acc_a==8 | pob_acc_b==8) & pob_acc==0

* Privación a nivel individual
 * Privación en consultas por enfermedad
 gen d_acc_a=(s17>7 & s17<12) if pob_acc_a==1
 replace d_acc_a=9 if s17==99 & pob_acc_a==1

* Privación en tratamiento de enfermedades AUGE-GES
 gen d_acc_b=(s30==2|s30==3|s30==5|s30==6|s30==7|s30==8|s30==10) if pob_acc_b==1
 replace d_acc_b=9 if s30==99 & pob_acc_b==1

* Privación en atención
 gen d_acc=(d_acc_a==1|d_acc_b==1) if pob_acc==1
 replace d_acc=9 if (d_acc_b==9 & s17==.)|(d_acc_b==9 & d_acc_a==9)|(d_acc_a==9 & s30==.)&
 pob_acc==1

*** 3.c. Indicadores de Trabajo y Seguridad Social**

* Indicador 7: Ocupacion
 * La población de referencia corresponde a las personas de 18 años o más.
 * Se considera carente a los hogares que al menos uno de sus integrantes se encuentra desocupado.

* Población de referencia
 gen pob_act=.
 replace pob_act=1 if edad>18
 replace pob_act=0 if edad<=18
 replace pob_act=1 if ((esc>12 & e6a==9) | (esc>13 & e6a==11)) & (edad>14 & edad<=18) & esc!=.
 replace pob_act=8 if esc==. & (edad>14 & edad<=18)

* Privación a nivel individual
 gen d_act=.
 replace d_act=1 if (activ==2) & pob_act==1
 replace d_act=0 if (activ==1 | activ==3) & pob_act==1
 replace d_act=9 if (activ==.) & pob_act==1

* Indicador 8: Seguridad Social
 * La población de referencia corresponde a las personas de 15 años o más que se encuentren ocupadas. Se excluye a trabajadores/as independientes con educación superior completa
 * Se considera carente a los hogares que al menos uno de sus integrantes no cotiza en el sistema previsional.

* Población de referencia
 * Se identifican a los independientes con educación superior completa

```
gen indsup=((e6a==15 | e6a==16 | e6a==17) & o15<3)
* Se genera a la población de referencia
gen pob_cot=.
replace pob_cot=1 if((edad>14) & activ==1)
replace pob_cot=0 if((edad<=14)|(activ==2|activ==3))
replace pob_cot=8 if pob_cot==.
```

* Privación a nivel individual
 replace o28=. if o28==9
 replace o29=. if o29==9

```
gen d_cot=0 if pob_cot==1
replace d_cot=1 if o28==2 & pob_cot==1
replace d_cot=1 if o28==1 & o29==7 & pob_cot==1
```

```
replace d_cot=. if o28==1 & o29==. & pob_cot==1
replace d_cot=. if o28==. & pob_cot==1
replace d_cot=0 if indsup==1 & pob_cot==1 & d_cot==1
replace d_cot=. if e6a==99 & o15<3 & d_cot==1
replace d_cot=9 if d_cot==. & pob_cot==1
```

* Indicador 9: Jubilación
 * La población de referencia son las personas en edad de jubilar.
 * Se considera carente a los hogares que al menos uno de sus integrantes no recibe pensión contributivo o no contributiva o que no reciben ingreso por arriendos, retiro de utilidades, dividendos o intereses.

* Población de referencia
 gen pob_jub=.
 replace pob_jub=1 if ((edad>64) & (sexo==1))
 replace pob_jub=0 if ((edad<=64) & (sexo==1))

```
replace pob_jub=1 if (edad>59) & sexo==2
replace pob_jub=0 if (edad<=59) & sexo==2
```

```
replace pob_jub=8 if pob_jub==.
```

*Ingresos por rentas, dividendos, utilidades, intereses.

*Ingresos por depósitos
 gen ydep=yah1

*Ingresos por dividendos por acciones
 g ydiv=yah2

*Ingresos por utilidades
g yretut=yrut

*Ingresos por arriendo de propiedades urbanas
g yrenturb=yre1

*Ingresos por arriendo de maquinarias
g yrentama=yama

*Ingresos por arriendo agricolas
g yrentagri=yre2

*Ingresos por arriendo de temporales
g yrentemp=yre3

*Suma de rentas, dividendos, utilidades, intereses.
egen yot=rowtotal(ydep ydiv yretut yrenturb yrentama yrentagri yrentemp)

* Privación a nivel individual

*** Se genera el ingreso por jubilaciones Contributiva: Pensión con APS, pensión, montepío, otra pensión
*/

*Pension con aporte APS
replace y26_2bm2=. if y26_2bm2==99

egen ypaps=rowtotal(y260201c y26_2bm2)

*Pensión
gen ypen=y2603c

*Montepío
gen ymonte=ymon

*Otra pensión
gen yotrp=yotrp

*Ingresos por jubilaciones
egen yj=rowtotal(ypaps ypen ymonte yotrp)

* Se genera la privación
gen d_jub=.
replace d_jub=0 if pob_jub==1
replace d_jub=1 if yj==0 & pob_jub==1

*Si recibe PBS es no carente (No contributivas)
replace d_jub=0 if y26_1a==1 & d_jub==1

* Si reciben Pensiones por leyes especiales de reparación son no carentes
replace d_jub=0 if yesp>0 & yesp<. & d_jub==1

* Si reciben ingresos de la propiedad son no carentes
replace d_jub=0 if yot>0 & yot<. & d_jub==1

*** 3.d. Vivienda y entorno**

* Indicador 10A: Hacinamiento

* La población de referencia corresponde a todos los hogares

* Se consideran carentes a los hogares que se encuentran hacinado (2,5 ó más personas por dormitorio de uso exclusivo).

* Población de Referencia

```
gen pob_hacina=(numper!=.)
replace pob_hacina=8 if numper==.
```

* Privación a nivel individual

```
gen d_hacina=.
replace d_hacina=(numper/v29a>=2.5) if tot_hog==1 & v29a>0 & v29a<99 & pob_hacina==1
replace d_hacina=(numper/v32a>=2.5) if tot_hog>1 & v32a>0 & v32a<99 & pob_hacina==1
replace d_hacina=1 if (tot_hog==1 & v29a==0)|(tot_hog>1 & v32a==0) & pob_hacina==1
replace d_hacina=9 if d_hacina==. & pob_hacina==1
```

* Indicador 10B: Estado de la Vivienda

* La población de referencia corresponde a todos los hogares

* Se consideran carentes a los hogares que residen en una vivienda precaria o en una vivienda con muros, techos y/o piso en mal estado.

* Población de Referencia

```
gen pob_estado=(numper!=.)
replace pob_estado=8 if numper==.
```

* Privación a nivel individual

```
gen d_estado=.
replace d_estado=1 if (v1==7|v1==10) & pob_estado==1
replace d_estado=1 if v1<=6 & (v3==3|v5==3|v7==3) & pob_estado==1
replace d_estado=1 if (v1==8|v1==9) & (v3==3|v5==3|v7==3) & pob_estado==1
replace d_estado=0 if v1<=6 & (v3<=2 & v5<=2 & v7<=2) & pob_estado==1
replace d_estado=0 if (v1==8|v1==9) & (v3<=2 & v5<=2 & v7<=2) & pob_estado==1

replace d_estado=9 if d_estado==. & pob_estado==1
```

* Indicador 10: Habitabilidad

* La población de referencia corresponde a todos los hogares. Un hogar es carente si se encuentra hacinado (2,5 ó más personas por dormitorio de uso

* exclusivo) o Reside en una vivienda precaria o en una vivienda con muros, techos y/o piso en mal estado.

* Población de referencia

```
gen pob_habitab=(numper!=.)
replace pob_habitab=8 if numper==.
```

* Privación a nivel individual

```

gen d_habitab=.
replace d_habitab=1 if (d_hacina==1 | d_estado==1) & pob_habitab==1
replace d_habitab=0 if (d_hacina==0 & d_estado==0) & pob_habitab==1
replace d_habitab=9 if d_habitab==. & pob_habitab==1

*****
* Indicador 11: Servicios Básicos
* La población de referencia corresponde a todos los hogares
* Se consideran carentes a los hogares que residen en una vivienda sin servicios sanitarios básicos (WC,
llave dentro de la vivienda y agua) según
* estándar urbano o rural.
*****

* Población de Referencia
gen pob_servbas=(numper!=.)
replace pob_servbas=8 if numper==.

* Componentes de la carencia
* Agua
gen agua=.
replace agua=1 if zona==1 & v23<=3 & pob_servbas==1
replace agua=1 if zona==2 & v23<=6 & pob_servbas==1
replace agua=2 if zona==1 & (v23==4|v23==5|v23==6|v23==7) & pob_servbas==1
replace agua=2 if zona==2 & (v23==7) & pob_servbas==1

* Disponibilidad del Agua
gen dis_agua=.
replace dis_agua=1 if v24==1 & pob_servbas==1
replace dis_agua=2 if v24==2|v24==3 & pob_servbas==1

* Eliminación de Excretas
gen elim_excretas=.
replace elim_excretas=1 if v25==1|v25==2 & pob_servbas==1
replace elim_excretas=2 if v25==3|v25==4|v25==5|v25==6|v25==7|v25==8 & pob_servbas==1

* Privación a nivel individual
gen d_servbas=.
replace d_servbas=0 if agua==1 & dis_agua==1 & elim_excretas==1 & pob_servbas==1
replace d_servbas=1 if agua==2 | dis_agua==2 | elim_excretas==2 & pob_servbas==1
replace d_servbas=9 if d_servbas==. & pob_servbas==1

*****
* Indicador 12: Entorno
* La población de referencia corresponde a todos los hogares.
*Un hogar se considera carente si su entorno no dispone de al menos un equipamiento básico y viven lejos
de su lugar de trabajo (1 hora de tiempo de traslado).
*También se considerarán carentes a aquellos hogares que residan en un entorno con 2 problemas de
contaminación (Contaminación del aire, de ríos, de agua de
* de la red pública o basura)
*****

* MEDIO AMBIENTE: durante los últimos 12 meses siempre presencia a lo menos DOS de las siguientes
situaciones, : contaminación del aire, ríos, agua
* pública o basura

```

```

* Situaciones de contaminación
*Aire
gen cont_aire=.
replace cont_aire=1 if v39b==4
replace cont_aire=0 if (v39b==1 | v39b==2 | v39b==3)
label variable cont_aire "Contaminación del aire"

*Contaminación de ríos, canales
gen cont_rios=.
replace cont_rios=1 if v39c==4
replace cont_rios=0 if (v39c==1 | v39c==2 | v39c==3)
label variable cont_rios "Contaminación de ríos, canales, esteros, lagos, tranques y embalses"

*Contaminación de agua de red pública
gen cont_agua=.
replace cont_agua=1 if v39d==4
replace cont_agua=0 if (v39d==1 | v39d==2 | v39d==3)
label variable cont_agua "Contaminación de agua de red pública"

*Contaminación por basura
gen cont_basura=.
replace cont_basura=1 if v39f==4
replace cont_basura=0 if (v39f==1 | v39f==2 | v39f==3)
label variable cont_basura "Acumulación de basura en calles, caminos, veredas o espacios públicos"

label define contaminacion 0 "Nunca, Pocas o Muchas veces" 1 "Siempre"

label values cont_aire contaminacion
label values cont_rios contaminacion
label values cont_agua contaminacion
label values cont_basura contaminacion

* Población de Referencia
gen pob_medio=(number!=.)
replace pob_medio=8 if number==.

*Suma de situaciones de contaminación
egen sum_con=rowtotal(cont_aire cont_rios cont_agua cont_basura)

* Privación a nivel individual
gen d_medio=.
replace d_medio=1 if (sum_con>=2) & pob_medio==1
replace d_medio=0 if (sum_con<2) & pob_medio==1
replace d_medio=9 if cont_aire==. & cont_rios==. & cont_agua==. & cont_basura==.

* EQUIPAMIENTO: no dispone de alguno de estos tres equipamientos mínimos: transporte público, centro
educacional, centro de salud

*Población de referencia
gen pob_equipo=(number!=.)
replace pob_equipo=8 if number==.

*Servicio de transporte público
gen d_eqtr=.
replace d_eqtr=1 if v37a==2
replace d_eqtr=0 if v37a==1
replace d_eqtr=. if v37a==9

```

```
*Servicio de centro educacional
gen d_eqce=.
replace d_eqce=1 if v37b==2
replace d_eqce=0 if v37b==1
replace d_eqce=. if v37b==9
```

```
*Servicio de centro de salud
gen d_eqcs=.
replace d_eqcs=1 if v37c==2
replace d_eqcs=0 if v37c==1
replace d_eqcs=. if v37c==9
```

```
*Acceso a equipos
```

```
gen d_equipo=.
replace d_equipo=0 if (d_eqtr==0 & d_eqce==0 & d_eqcs==0) & pob_equipo==1
replace d_equipo=1 if (d_eqtr==1 | d_eqce==1 | d_eqcs==1) & pob_equipo==1
replace d_equipo=9 if d_equipo==. & pob_equipo==1
```

```
*****
* TIEMPO DE TRASLADO: tiempo de traslado al trabajo en transporte público o particular no motorizado
* igual o mayor a 60 min.
*****
```

```
* Población de referencia
gen oc=.
replace oc=1 if ((o1==1 | o2==1 | o3==1) & (edad>14))
replace oc=0 if ((o3==2) & (o6==1|o6==2) & (edad>14))
replace oc=0 if edad<=14
label var oc "Ocupados"
label define oc 1 "Si" 0 "No"
label values oc oc
```

```
g pob_tiempo=(oc==1)
replace pob_tiempo=8 if oc==.
```

```
* Privación a nivel individual
gen d_tiempo=0 if (o25a_hr<1 & o25a_hr!=.) & pob_tiempo==1
replace d_tiempo=1 if (o25a_hr>=1 & o25a_hr!=.) & pob_tiempo==1
replace d_tiempo=9 if o25a_hr==. & pob_tiempo==1
```

```
*Medio de transporte
replace d_tiempo=0 if (o25c==2 | o25c==5) & d_tiempo==1 & pob_tiempo==1
replace d_tiempo=9 if (o25c==.) & d_tiempo==1 & pob_tiempo==1
```

```
*CARENCIA EN TIEMPO DE TRASLADO POR HOGAR
```

```
bys folio: egen hh_oc=max(oc)
bys folio: egen d_hh_tiempo=max(d_tiempo) if hh_oc==1
```

```
*Accesibilidad
```

```
gen pob_accesi=1
```

```
gen d_accesi=.
```

```
replace d_accesi=1 if (d_hh_tiempo==1 & d_equipo==1) & hh_oc==1
replace d_accesi=0 if (d_hh_tiempo==0 | d_equipo==0) & hh_oc==1
```

```
replace d_accesi=1 if (d_equipo==1) & hh_oc==0
replace d_accesi=0 if (d_equipo==0) & hh_oc==0
replace d_accesi=9 if d_accesi==.
```

```
*****
*Entorno
*****
```

```
* Población de referencia
gen pob_entorno=(numper!=.)
replace pob_entorno=8 if (numper==.)
```

```
* Privación a nivel individual
gen d_entorno=.
replace d_entorno=0 if d_medio==0 & d_accesi==0 & pob_entorno==1
replace d_entorno=1 if d_medio==1 | d_accesi==1 & pob_entorno==1
replace d_entorno=9 if d_entorno==.
```

```
*****
* 3.e. Redes y Cohesión social
*****
```

```
*****
* Indicador 13: Apoyo y Participación Social
/* Un hogar es carente si no dispone de alguna persona fuera del hogar que pueda prestar apoyo para 8
situaciones relevantes ,
ninguno de sus miembros de 14 y más años participa en alguna organización social o grupo organizado y
ninguno de sus miembros de 18 y más años
que se encuentran ocupados está afiliado a alguna organización sindical, gremial o profesional relacionada
con su trabajo */
*****
```

```
*****
* Apoyo Social- Un hogar es carente si no dispone de alguna persona fuera del hogar para las 8
situaciones
*****
```

```
*Población de referencia

gen pob_hapoyo=(numper!=.)
replace pob_hapoyo=8 if numper==.
```

* Se identifican las categorías en las cuales se considera carente para cada situación

```
gen apoyo_a=.
replace apoyo_a=0 if r7a==4
replace apoyo_a=0 if r7a==2
replace apoyo_a=1 if r7a==1
replace apoyo_a=1 if r7a==3
```

```
gen apoyo_c=.
replace apoyo_c=0 if r7c==4
replace apoyo_c=0 if r7c==2
replace apoyo_c=1 if r7c==1
```

```
replace apoyo_c=1 if r7c==3

gen apoyo_d=.
replace apoyo_d=0 if r7d==4
replace apoyo_d=0 if r7d==2
replace apoyo_d=1 if r7d==1
replace apoyo_d=1 if r7d==3

gen apoyo_e=.
replace apoyo_e=0 if r7e==4
replace apoyo_e=0 if r7e==2
replace apoyo_e=1 if r7e==1
replace apoyo_e=1 if r7e==3

gen apoyo_f=.
replace apoyo_f=0 if r7f==4
replace apoyo_f=0 if r7f==2
replace apoyo_f=1 if r7f==1
replace apoyo_f=1 if r7f==3

gen apoyo_g=.
replace apoyo_g=0 if r7g==4
replace apoyo_g=0 if r7g==2
replace apoyo_g=1 if r7g==1
replace apoyo_g=1 if r7g==3

gen apoyo_h=.
replace apoyo_h=0 if r7h==4
replace apoyo_h=0 if r7h==2
replace apoyo_h=1 if r7h==1
replace apoyo_h=1 if r7h==3

gen apoyo_i=.
replace apoyo_i=0 if r7i==4
replace apoyo_i=0 if r7i==2
replace apoyo_i=1 if r7i==1
replace apoyo_i=1 if r7i==3

label define apoyo_lbl 0 "No conoce fuera del hogar" 1 "Conoce fuera del hogar"
label value apoyo_a apoyo_lbl
label value apoyo_c apoyo_lbl
label value apoyo_d apoyo_lbl
label value apoyo_e apoyo_lbl
label value apoyo_f apoyo_lbl
label value apoyo_g apoyo_lbl
label value apoyo_h apoyo_lbl
label value apoyo_i apoyo_lbl

label variable apoyo_a "Enfermedad"
label variable apoyo_c "Vehiculo"
label variable apoyo_d "Dinero"
label variable apoyo_e "Trámites"
label variable apoyo_f "Tecnologías"
label variable apoyo_g "Reparaciones"
label variable apoyo_h "Trabajo"
label variable apoyo_i "Problemas personales"
```

```

*** Agregacion por hogar
local apoyo a c d e f g h i
foreach var in `apoyo' {
bys folio: egen hapoyo_`var'=max(apoyo_`var')
}

* Genera suma que permite identificas carencia
egen c_apoyo=rsum(hapoyo_a hapoyo_c hapoyo_d hapoyo_e hapoyo_f hapoyo_g hapoyo_h hapoyo_i)
egen c_apoyo_miss=rmiss (hapoyo_a hapoyo_c hapoyo_d hapoyo_e hapoyo_f hapoyo_g hapoyo_h
hapoyo_i)
replace c_apoyo=. if c_apoyo_miss>=1 & c_apoyo==0
label variable c_apoyo "Indicador apoyos sociales"
label variable c_apoyo_miss "Contador de variables perdidas en apoyos"

* Privación si para todas las situaciones no cuenta con alguna persona a fuera del hogar.
gen d_hapoyo=.
replace d_hapoyo=1 if c_apoyo==0 & pob_hapoyo==1 & c_apoyo!=. & c_apoyo_miss==0
replace d_hapoyo=0 if (c_apoyo>=1) & pob_hapoyo==1 & c_apoyo!=.
replace d_hapoyo=9 if d_hapoyo==. & pob_hapoyo==1

*****
* Participación social
/*Un hogar se considera carente si ninguno de sus miembros mayores de 14 años participa en una
organización social (alternativas módulo Residentes) y
ninguno miembro mayor de 18 años está afiliado a organización sindical o gremial en el trabajo
(alternativas de modulo de Trabajo). */
*****

*Población de referencia
g pob_part=(edad>=14)

*Participación en organizaciones

*Organizaciones sociales
g p_orgs=.
replace p_orgs=1 if (r6<=14) & pob_part==1
replace p_orgs=0 if (r6==15) & pob_part==1

*Organizaciones del trabajo
g p_orgt=.
replace p_orgt=1 if ((o24a==1 | o24b==1 | o24c==1 | o24d==1) & activ==1 & edad>=18)
replace p_orgt=0 if ((o24a==2 & o24b==2 & o24c==2 & o24d==2) & activ==1 & edad>=18)

label define orglbl 0 "No participa" 1 "Si participa"
label value p_orgs orglbl
label value p_orgt orglbl

*Ambas organizaciones.
g d_part=.

*Para población mayor de 14 años y menor de 18 años, sólo pueden participar en organizaciones sociales

replace d_part=1 if p_orgs==0 & pob_part==1 & edad<18
replace d_part=0 if p_orgs==1 & pob_part==1 & edad<18
replace d_part=9 if p_orgs==. & pob_part==1 & edad<18

*Población mayor de 18 años

```

*Población mayor de 18 años e INACTIVA, sólo pueden participar en organizaciones sociales

```
replace d_part=1 if p_orgs==0 & (activ==2|activ==3) & edad>=18
replace d_part=0 if p_orgs==1 & (activ==2|activ==3) & edad>=18
replace d_part=9 if p_orgs==. & (activ==2|activ==3) & edad>=18
```

*Para población mayor de 18 años y ACTIVA, pueden participar en organizaciones sociales Y organizaciones laborales

```
replace d_part=1 if (p_orgs==0 & p_orgt==0) & activ==1 & pob_part==1 & edad>=18
replace d_part=0 if (p_orgs==1 | p_orgt==1) & activ==1 & pob_part==1 & edad>=18
replace d_part=9 if ((p_orgs==0 & p_orgt==.)|(p_orgs==. & p_orgt==.)) | (p_orgs==. & p_orgt==0) &
activ==1 & pob_part==1 & edad>=18
```

label define d_part 0 "No carente" 1 "Carente"

label value d_part d_part

* Apoyo y Participación Social

*Población de referencia

g pob_appart=(numper!=.)

g d_appart=.

```
replace d_appart=1 if (d_hapoyo==1) & edad<14
replace d_appart=0 if (d_hapoyo==0) & edad<14
```

```
replace d_appart=1 if (d_part==1 & d_hapoyo==1) & pob_appart==1
replace d_appart=0 if (d_part=0 | d_hapoyo=0) & pob_appart==1
replace d_appart=9 if d_appart==. & (d_hapoyo==9 | d_part==9) & pob_appart==1
```

* Indicador 14: TRATO IGUALITARIO

/* Un hogar es carente si no dispone de alguna persona fuera del hogar que pueda prestar apoyo para 8 situaciones relevantes ,
ninguno de sus miembros de 14 y más años participa en alguna organización social o grupo organizado y ninguno de sus miembros de 18 y más años
que se encuentran ocupados está afiliado a alguna organización sindical, gremial o profesional relacionada con su trabajo */

* Población de Referencia

gen pob_tsocial=(numper!=.)

replace pob_tsocial=8 if numper==.

* Privación a nivel individual

gen d_tsocial=.

```
replace d_tsocial=1 if (r8a==1 | r8b==1 | r8c==1 | r8d==1 | r8e==1 | r8f==1 | r8g==1 | r8h==1 | r8i==1 | r8j==1 | r8k==1 | r8l==1 | r8m==1 | r8n==1 | r8o==1 | r8p==1 | r8q==1) & pob_tsocial==1
```

```
replace d_tsocial=0 if (r8a==2 & r8b==2 & r8c==2 & r8d==2 & r8e==2 & r8f==2 & r8g==2 & r8h==2 & r8i==2 & r8j==2 & r8k==2 & r8l==2 & r8m==2 & r8n==2 & r8o==2 & r8p==2 & r8q==2) & (r8s==1) & pob_tsocial==1
```

```
replace d_tsocial=0 if (r8s==1) & pob_tsocial==1
```

```
replace d_tsocial=0 if (r8r==1) & d_tsocial==. & pob_tsocial==1
```

```

*****
* Indicador 15:SEGURIDAD
/*Un hogar se considera carente si declara haber presenciado durante el último mes a lo menos una de las
siguientes situaciones con
frecuencia "siempre": i)Tráfico de drogas; ii) Balaceras o disparos. */
*****

g trafico=.
replace trafico=1 if v38c==4
replace trafico=0 if (v38c==1 | v38c==2 | v38c==3)

g balacera=.
replace balacera=1 if v38e==4
replace balacera=0 if (v38e==1 | v38e==2 | v38e==3)

label define seguridad 0 "Nunca, Pocas o Muchas veces" 1 "Siempre"

label values trafico seguridad
label values balacera seguridad

*Población de referencia
gen pob_seg=(numper!=.)
replace pob_seg=8 if pob_seg==.

*Indicador (alguna de las situaciones críticas ocurre siempre)
gen d_seg=.
replace d_seg=0 if pob_seg==1
replace d_seg=1 if (trafico==1 | balacera==1) & pob_seg==1
replace d_seg=9 if d_seg=. & pob_seg==1

*****
*****
*4. AGREGACIÓN DE INCIDENCIA POR HOGARES
*****
*****

local miss asis rez esc mal prevs acc act cot jub hacina estado habitab servbas entorno appart tsocial seg
accesi medio
foreach var in `miss' {
g pob_`var'_c=pob_`var'
replace pob_`var'_c=0 if (pob_`var'_c==8 | pob_`var'_c==.)
g d_`var'_c=d_`var'
replace d_`var'_c=. if (d_`var'_c==9 | d_`var'_c==.)
}

local pob asis rez esc mal prevs acc act cot jub habitab servbas entorno tsocial seg hacina estado accesi
medio

foreach var in `pob' {
bys folio: egen hh_pob_`var'=max(pob_`var'_c)
bys folio: egen hh_d_`var'=max(d_`var'_c)
replace hh_d_`var'=0 if hh_pob_`var'==0
label var d_`var'_c "Individuos carentes en `var' de población relevante"
}

```

```

local pob appart
foreach var in `pob' {
  bys folio: egen hh_pob_`var'=max(pob_`var'_c)
  bys folio: egen hh_d_`var'=min(d_`var'_c)
  replace hh_d_`var'=0 if hh_pob_`var'==0
  label var d_`var'_c "Individuos carentes en `var' de población relevante"
}

```

```

label var hh_d_asis "hogar carente en asistencia"
label var hh_d_rez "hogar carente en rezago escolar"
label var hh_d_esc "hogar carente en escolaridad"
label var hh_d_mal "hogar carente en malnutrición en niños/as"
label var hh_d_prevs "hogar carente en adscripción a sistema de salud"
label var hh_d_acc "hogar carente en atención"
label var hh_d_act "hogar carente en ocupación"
label var hh_d_cot "hogar carente en seguridad social"
label var hh_d_jub "hogar carente en jubilaciones"
label var hh_d_habitab "hogar carente en habitabilidad"
label var hh_d_hacina "hogar carente en hacinamiento"
label var hh_d_estado "hogar carente en estado de la vivienda"
label var hh_d_servbas "hogar carente en servicios básicos"
label var hh_d_entorno "hogar carente en entorno"
label var hh_d_accesi "hogar carente en accesibilidad"
label var hh_d_medio "hogar carente en medio ambiente"
label var hh_d_appart "hogar carente en apoyo y participación social"
label var hh_d_tsocial "hogar carente en trato igualitario"
label var hh_d_seg "hogar carente en seguridad"

```

```

*****
*****
*5. IDENTIFICACIÓN Y AGREGACIÓN DE MEDIDA DE POBREZA CON ENTORNO Y REDES
*****
*****

```

```

*****
*5.1. Ponderación
*****

```

*PESO DE DIMENSIONES TRADICIONALES

local pob asis rez esc mal prevs acc act cot jub habitab servbas entorno

```

foreach var in `pob' {
  * Peso para cada indicador
  gen pp_`var'= 0.225/3
  * Carencia ponderada
  gen w_hh_d_`var'= hh_d_`var'* pp_`var'
  *label var hh_d_`var' "Hogar carente en `var'"
  label var pp_`var' "Ponderador `var'"
  label var w_hh_d_`var' "Carencia ponderada de `var'"
}
*

```

*PESO DE DIMENSIÓN DE COHESIÓN SOCIAL

```
local pob appart tsocial seg
```

```
foreach var in `pob' {
* Peso para cada indicador
gen pp_`var'= 0.1/3
* Carencia ponderada
gen w_hh_d_`var'= hh_d_`var'* pp_`var'
label var pp_`var' "Ponderador `var'"
label var w_hh_d_`var' "Carencia ponderada de `var'"
}
*
```

```
*****
```

```
*5.2 Vector de conteo
```

```
*****
```

```
* El siguiente comando suma todos los indicadores ponderados
```

```
egen ci=rsum(w_hh_d_*)
label var ci "Vector de Conteo"
```

```
* Se identifican hogares excluidos por no contar con información para todos los indicadores
```

```
egen missing=rmiss(w_hh_d_*)
gen dmissing=(missing!=0)
label var missing "Conteo de Missing"
```

```
*****
```

```
* 5.3. Identificación y Agregación
```

```
*****
```

```
*UMBRAL K=22.5%
```

```
gen h_k=(ci>=0.225) if dmissing!=1
label var h_k "situación de pobreza multidimensional con entorno y redes (5 dimensiones)"
gen a_k=ci if h_k==1
label var a_k "Intensidad de la pobreza multidimensional con entorno y redes (5 dimensiones)"
gen m_k=0 if dmissing!=1
replace m_k=ci if h_k==1
label var m_k "Tasa de recuento ajustado de pobreza multidimensional con entorno y redes (5 dimensiones)"
```

```
*****
```

```
*****
```

```
*6. IDENTIFICACIÓN Y AGREGACIÓN DE MEDIDA DE POBREZA DE 4 DIMENSIONES
```

```
*****
```

```
*****
```

```
*****
```

```
*6.1. Ponderación
```

```
*****
```

```
*PESO DE DIMENSIONES TRADICIONALES
```

```
local pob asis rez esc mal prevs acc act cot jub hacina estado servbas
```

```
foreach var in `pob' {
* Peso para cada indicador
gen pp_`var'_ant= 0.25/3
* Carencia ponderada
```

```
gen w_hh_d_`var'_ant= hh_d_`var'* pp_`var'_ant
label var pp_`var'_ant "Ponderador `var'"
label var w_hh_d_`var'_ant "Carencia ponderada de `var'"
}
*
```

*6.2. Vector de conteo

* El siguiente comando suma todos los indicadores ponderados

```
egen ci_ant=rsum(w_hh_d*_ant)
label var ci_ant "Vector de Conteo"
```

* Se identifican hogares excluidos por no contar con información para todos los indicadores

```
egen missing_ant=rmiss(w_hh_d*_ant)
gen dmissing_ant=(missing_ant!=0)
label var missing_ant "Conteo de Missing"
```

*6.3 Identificación y Agregación

*UMBRALES K=25%

```
gen h_ant=(ci_ant>=0.25) if dmissing_ant!=1
label var h_ant "situación de pobreza multidimensional (4 dimensiones)"
gen a_ant=ci_ant if h_ant==1
label var a_ant "Intensidad de pobreza multidimensional (4 dimensiones)"
gen m_ant=0 if dmissing_ant!=1
replace m_ant=ci_ant if h_ant==1
label var m_ant "Tasa de recuento ajustado de pobreza multidimensional (4 dimensiones)"
```

*Se cambia nombre de la variable de incidencia para cada una de las medidas

```
rename h_k pobreza_multi_5d
label define pobreza_multi 0 "no pobre" 1 "pobre"
label values pobreza_multi_5d pobreza_multi
```

```
rename h_ant pobreza_multi_4d
label values pobreza_multi_4d pobreza_multi
```